/*
 * Flashlib - QB9 common Flash library
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

package com.qb9.flashlib.display.spritesheet
{
	import com.qb9.flashlib.interfaces.IDisposable;
	
	import flash.display.BitmapData;

	public class SpriteSequence implements IDisposable, SpriteSource
	{
		private var sheet:SpriteSheet;
		private var map:Array;
		
		public function SpriteSequence(sheet:SpriteSheet, map:Array=null)
		{
			this.sheet = sheet;
			this.map = map || [];
		}

		public function add(index:uint, amount:uint=1):void
		{
			while (amount--)
				map.push(index);
		}

		public function get length():uint
		{
			return map.length;
		}
		
		public function getSprite(i:uint):BitmapData
		{
			return i in map ? sheet.getSprite(map[i]) : null;
		}
		
		public function dispose():void
		{
			sheet = null;
			map = null;
		}
	}
}